

A SAS® Macro Approach to Assign CTCAE Grades to Laboratory Adverse Experiences

Mei Dey, Accenture, Raleigh, NC
Lisa Pyle, Accenture, Philadelphia, PA

ABSTRACT

The Common Terminology Criteria for Adverse Events (CTCAE) published by the National Cancer Institute (NCI) is widely used in the oncology therapeutic area and provides a severity grading scale for adverse experiences. This guideline describes the severity grading for clinical adverse experiences as well as certain laboratory results. Typically, CTCAE grading is directly collected from the site on the adverse experience case report form. However, this may not be the case for laboratory results. Oftentimes, only the lab result/unit and normal ranges are collected without any indication of toxicity severity.

This paper discusses a SAS utility macro designed to specifically apply toxicity grading to selected laboratory results based on the CTCAE guidance and the challenges surrounding the implementation. This macro can be utilized on laboratory results collected from either local or central laboratories once all units are converted to standardized measures. Lastly, this utility macro was originally designed to use Version 3.0 CTCAE to populate the LBTOXGR column in the LB SDTM domain per CDISC standards but it could easily be modified to reflect the more current CTCAE Version 4.0.

CTCAE GUIDELINE

The NCI CTCAE is a descriptive terminology applied to Adverse Event (AE) for reporting in clinical trials. It is designed as an instrument to be used to document AEs identified through a combination of clinical and laboratory evaluation.

A categorical grading scale is described in the guideline for each AE term for the purposes of regulatory reporting to the NCI. A CTCAE grade refers to the severity of the AE. It displays Grade 1 through Grade 5 based on this general guideline:

Table 1 NCI Grading Severity Scale

1 = Mild Adverse Event
2 = Moderate Adverse Event
3 = Severe Adverse Event
4 = Life-threatening Adverse Event
5 = Death related to Adverse Event

For clinical AE evaluation in oncology trials, CTCAE grades are collected as a question on the AE case report form (CRF) by the site investigators using the NCI CTCAE guideline as the basis for their assessment.

Unlike clinical AE evaluation where a descriptive terminology is described in the guidelines, for lab AEs, the guideline references an algorithm to be applied to selected laboratory results, laboratory units and normal ranges. Since this algorithm is not straightforward and requires some derivation, lab CRFs typically omits the collection of toxicity grade to avoid human error in the collection of toxicity grade. Instead, the lab CRFs only collect the lab result, lab unit and lab normal ranges without any indication of toxicity severity.

While most clinical trials report laboratory results based solely on the normal range criteria where the result is lower or higher than what is considered “normal”, oncology trials must report on the CTCAE grade for abnormal lab tests. As a result, clinical trial sponsors are challenged to develop a tool to apply the CTCAE grading for both local and central lab results to ensure consistency.

MACRO DESIGN

This paper focuses on designing a SAS utility macro to apply the CTCAE toxicity grading to selected laboratory results.

FUNCTIONAL DESCRIPTION

This SAS macro assigns CTCAE grades based on the CTCAE version specified in a pre-defined metadata file using the algorithms described in the corresponding guideline.

The macro produces a derivation result that populates the LBTOXGR column in the SDTM LB domain for a pre-specified list of laboratory tests/unit combinations listed in the required metadata file. The result is a character value of numeric integer with values 1, 2, 3, 4, NULL or error code E1 thru E7. A NULL value in the LBTOXGR field indicates a 'normal' result and grade 5 is not applicable.

This derivation has the option to execute on all lab tests in the SDTM LB domain including local, central and supplemental lab tests or some subset of labs as specified by the user.

This derivation has the option to completely replace all the LBTOXGR existing results (if present in the input LB dataset) and apply the macro derivations to the pre-specified test/units or it will have the ability to apply derivations only to the pre-specified list of test/units in the metadata as specified by the user.

For lab tests/units not listed in the metadata table, the LBTOXGR derivation is populated with NULL unless specified otherwise by the user (see parameter section - Apply_toxgrade)

In addition, since NULL is a valid result, for error exception handling a value of 'E' is used to indicate derivation failures.

Lastly, the derivation populates the LBTOX variable in the LB SDTM domain. This column will be populated with the text "NCI CTCAE vX.X" as a constant for all gradable laboratory tests specified in metadata and not excluded from &where parameter.

INPUT REQUIREMENTS

1. Lab dataset

The input lab dataset for this macro is expected to be in SDTM format (i.e. LB SDTM DOMAIN or similar) and it must include the minimum required variables listed below.

Table 2 SDTM Required Variables

LB.LBTESTCD	Lab Test Code
LB.LBCONRLO	Reference Range Lower Limit Preferred Units
LB.LBCONRHI	Reference Range Upper Limit Preferred Units
LB.LBCONN	Numeric Result in Preferred Units
LB.LBCONU	Preferred Units

It is assumed that all laboratory results and normal reference ranges have been converted into preferred or standard units (e.g. conventional units (US)) and that any unit of measurement conversion has been applied. Further, if any percent to absolute value conversions are required, they must also be applied prior to implementing this macro.

In addition, lab tests specimens (LBSPEC) for pre-specified tests in the input dataset are assumed to be in sync with the NCI CTCAE guidance document.

2. Metadata file

Since the NCI CTCAE guideline does not change often, a metadata file is developed from the CTCAE guideline with the appropriate laboratory derivations to facilitate the CTCAE assignment to maximize efficiency and reusability.

The metadata file provides the details for "gradable" lab tests. A gradable lab test is defined as a lab test that can solely be graded based on collected lab results as defined in the guidance document. For example, per the NCI CTCAE v3.0 guidance document, Uric Acid Grade 1 is described as a uric acid result that is greater than the upper limit of normal and less than 10 mg/dL *without physiologic consequences*. Since the grading of this laboratory test requires some clinical assessment beyond the collected lab results to apply a grading, this lab is excluded from the metadata and is therefore, considered not "gradable" using a macro solution. On the other hand, the grading for glucose relies solely on the result and the upper limit of normal to evaluate severity. As a result, glucose is a "gradable" test and resides in the metadata file along with the parameters and derivation algorithm.

Table 3 Sample Macro Metadata File Contents

TESTCD	CONVENTIONAL_UNIT	GRADE	LOWER_COMPARISON	LOWER_VALUE	LOWER_MULTIPLIER	HIGHER_COMPARISON	HIGHER_VALUE	HIGHER_MULTIPLIER	NCI_VERSION
CA	mg/dL	1	<=	8	1	>	1	LBCONRLO	NCI CTCAE v3.0
CA	mg/dL	1	<	1	LBCONRHI	>=	11.5	1	NCI CTCAE v3.0
CA	mg/dL	2	<	11.5	1	>=	12.5	1	NCI CTCAE v3.0
CA	mg/dL	2	<=	7	1	>	8	1	NCI CTCAE v3.0
CA	mg/dL	3	<	12.5	1	>=	13.5	1	NCI CTCAE v3.0
CA	mg/dL	3	<=	6	1	>	7	1	NCI CTCAE v3.0
CA	mg/dL	4				>	6	1	NCI CTCAE v3.0
CA	mg/dL	4	<	13.5	1				NCI CTCAE v3.0
CREAT	mg/dL	1	<	1	LBCONRHI	>=	1.5	LBCONRHI	NCI CTCAE v3.0
CREAT	mg/dL	2	<	1.5	LBCONRHI	>=	3	LBCONRHI	NCI CTCAE v3.0
CREAT	mg/dL	3	<	3	LBCONRHI	>=	6	LBCONRHI	NCI CTCAE v3.0
CREAT	mg/dL	4	<	6	LBCONRHI				NCI CTCAE v3.0
CHOL	mg/dL	1	<	1	LBCONRHI	>=	300	1	NCI CTCAE v3.0
CHOL	mg/dL	2	<	300	1	>=	400	1	NCI CTCAE v3.0
CHOL	mg/dL	3	<	400	1	>=	500	1	NCI CTCAE v3.0
CHOL	mg/dL	4	<	500	1				NCI CTCAE v3.0

MACRO PARAMETERS

The following input parameters are chosen and designed to handle multiple input scenarios and to offer maximum flexibility and functionality.

Table 4 Macro Parameter Descriptions

Parameter	Description	Default	Required	Valid Values & Comments
Input_dataset	Input Laboratory Dataset	xxx.lb	Y	This is the input laboratory SDTM lab data set with the minimum required variables listed in the Inputs section above.
Output_dataset	Output Laboratory Dataset	xxx.lb	N	The default output data set is the same as the input_dataset unless otherwise specified by the user.
Where	Selects rows of data in LB domain to implement toxicity grading derivation.		N	LBGRPID in ("LABS","SLABS") Enables a user to apply the derivation to some subset of vendor labs or all labs. ex) where LBGRPID in ("LABS","SLABS") The default option is that the derivation executes on all lab test observations.
Input_metadata	BR1000 metadata data set	xxx.metatdata	Y	Metadata that details the grading derivations for specific lab tests/units.
Version	CTCAE lab grading version	3.0	N	3.0 or 4.0 Filters metadata in Input_metadata for v3.0 or v4.0 derivations.
Apply_toxgrade	Specifies whether to apply derivation from scratch OR whether apply it to pre-specified tests/units only.	1	Y	1 or 2 1 - BR completely replaces any existing LBTOXGR data. 2 - BR overrides LBTOXGR only for pre-specified tests/units in metadata. (For ex) if URATE is graded but not in metadata no override is applied. Whereas, if ALB is graded and ALB is in metadata, then the BR will override the result.
Print_ErrRpt	Outputs an Error Report in Excel of laboratory tests that can not be graded due to missing derivation information.	Y	N	For gradable tests, if Y produces a listing of ERROR codes for missing required derivation inputs. Error Codes: E1 thru E7
Debug		N	N	Y or N

MACRO CODE DESCRIPTION & SAMPLES

1. The first step of the macro is to read in the metadata file and perform the following steps.
 - a. Filter metadata for rows where the NCI_version is &version
 - b. Get an array of unique test codes.
 - c. Determine the maximum number of unique test codes.

- d. Determine the maximum number of derivations per test/unit combination. Read in each metadata spreadsheet row . (Note: each row represents the components for one derivation per test/unit combination.)

```

%*-----
----*;
%*- Step 01: Read in CTCAE metadata file.
-*.
%*-----
----*;
proc sql noprint;
  create table _inmeta as
  select *
  from &input_metadata
  where nci_version = "NCI CTCAE v&version";

  select count(distinct testcd)
  into :tottest
  from _inmeta;

  select distinct nci_version
  into :ver
  from _inmeta;

  select distinct testcd, conventional_unit, count(*)
  into :test1 - :test%left(&tottest),
       :unit1 - :unit%left(&tottest),
       :der1 - :der%left(&tottest)
  from _inmeta
  group by testcd;
quit;

%do i=1 %to &tottest;

data _test;
  set _inmeta end=final;
  where testcd = "&&test&i";
run;

data _null_;
  set _test;
  call symput("grade&i"||left(_n_), trim(left(grade)));
  call symput("lc&i"||left(_n_), trim(left(lower_comparison)));
  call symput("lv&i"||left(_n_), trim(left(lower_value)));
  call symput("lm&i"||left(_n_), trim(left(lower_multiplier)));
  call symput("hc&i"||left(_n_), trim(left(higher_comparison)));
  call symput("hv&i"||left(_n_), trim(left(higher_value)));
  call symput("hm&i"||left(_n_), trim(left(higher_multiplier)));
run;
%end;

```

2. Next the macro applies the CTCAE grade algorithm to the input lab dataset.

For each test/unit combination in the input metadata file, &inmeta, where the result is outside the normal range

LBCONN <= LBCONRLO OR LBCONN >= LBCONRHI perform the derivations as noted in the metadata file, input_metadata, as follows:

- I. If lower_comparison is missing then
LBTOXGR = GRADE WHEN
higher_value * higher_multiplier higher_comparison LBCONN is true

- II. If higher_comparison is missing then
 LBTOXGR = GRADE WHEN
 lower_value * lower_multiplier lower_comparison LBCONN is true
- III. If lower_comparison ne missing and the higher_comparison ne missing then
- a. If lower_value * lower_multiplier = higher_value * higher_multiplier and lower_value * lower_multiplier = Result (LBCONN) then assign it to that grade.
 - b. else LBTOXGR = GRADE when lower_value * lower_multiplier lower_comparison LBCONN is true AND higher_value * higher_multiplier higher_comparison LBCONN is true

```

%*-----*
----*;
%*- Step 02: Apply grading to laboratory data.
-*.
%*-----*
----*;

data &output_dataset;
    set _temp;
    length _LBTOXGR $3;

/* Apply the toxgrade derivation to all tests */
%do i=1 %to &totttest;

/* apply toxgrade derivation to the flagged observations */
if _apply=1
    and lbtestcd = "&&test&I"
    and lbconu = "&&unit&I"
    and lbconn ne .
    and (lbconn <= lbconrlo or lbconn => lbconrhi) then do;

    /* for each derivation per lab test */

    %do j=1 %to &&der&i;
    %if "&&lc&i&j" eq " " %then %do;
        if &&hv&i&j * &&hm&i&j &&hc&i&j LBCONN
            then _LBTOXGR=trim(left(&&grade&i&j));
    %end;

    %if "&&hc&i&j" eq " " %then %do;
        if &&lv&i&j * &&lm&i&j &&lc&i&j LBCONN
            then _LBTOXGR=trim(left(&&grade&i&j));
    %end;

    %if "&&lc&i&j" ne " " and "&&hc&i&j" ne " " %then %do;
        if (&&lv&i&j * &&lm&i&j ne &&hv&i&j * &&hm&i&j) then do;
            if (&&lv&i&j * &&lm&i&j &&lc&i&j LBCONN) and
                (&&hv&i&j * &&hm&i&j &&hc&i&j LBCONN)
                then _LBTOXGR=trim(left(&&grade&i&j));
        end;
    %end;

    %if "&&lc&i&j" ne " " and "&&hc&i&j" ne " " %then %do;
        if ((&&lv&i&j * &&lm&i&j EQ &&hv&i&j * &&hm&i&j) and
            (&&lv&i&j * &&lm&i&j eq LBCONN))
            then _LBTOXGR=trim(left(&&grade&i&j));
    %end;

    %end;

end;
%end;

```

```
run;
```

3. A sample macro call

```
%implement0br1000(input_dataset=lb,
    output_dataset=labsx,
    where=,
    input_metadata=source.metadata,
    version=3.0,
    print_errrpt=y,
    input_dataset2=,
    debug=Y);
```

ERROR HANDLING

This macro also has error checking codes to help user understand data issues that can prevent macros from functionally correctly.

1. Confirm that the input data set has the required columns for the derivation: LB.LBTESTCD, LB.LBCONU, LB.LBCONRLO, LB.LBCONRHI and LB.LBCONN. If any required variable is missing then issue error message (e.g E4) and abort. The following report error codes detail missing derivation components that must be non-missing in order for the derivation to successfully execute.

Table 5 Macro Error Handling Report Code Descriptions

E1	Only the laboratory value LBCONN is missing.
E2	Only the laboratory unit LBCONU is missing.
E3	Only the laboratory normal ranges LBCONRLO and LBCONRHI are missing.
E4	The laboratory value LBCONN and the laboratory unit LBCONU are missing.
E5	The laboratory value LBCONN and the laboratory normal ranges LBCONRLO and LBCONRHI are missing.
E6	The laboratory unit LBCONU and the laboratory normal ranges LBCONRLO and LBCONRHI are missing.
E7	The laboratory value LBCONN, the laboratory unit LBCONU and the laboratory normal ranges LBCONRLO and LBCONRHI are missing.

2. If the LBTOXGR variable is non-missing and the LBTOXGR has non-missing values and apply_toxgrade = 1 then the macro issues a warning message in the log file as follows:

Message: This BR will overwrite all existing toxicity grades.

This is to alert the macro user that the lab file already contains CTCAE grading and confirms if the user really wants to override all existing toxicity grades by executing this macro.

3. A similar warning applies when apply_toxgrade = 2 is selected and the LBTOXGR variable is non-missing and the LBTOXGR has non-missing values.

Message: This BR may overwrite some existing toxicity grades.

MACRO OUTPUTS

The macro generates a new LB dataset with additional variables of LBTOXGR (CTCAE grades) and LBTOX (indicating CTCAE version applied)

If parameter Print_ErrRpt is selected, the macro can also produce a listing of lab tests with ERROR codes indicating missing required derivation inputs (E1 thru E7 above). This error report can help data management to clean up the data further prior to macro execution.

CHALLENGES OF IMPLEMENTATION

Bi-directional Lab Tests

Lab tests are, in general, graded by the degree to which the result is out of range high or when the result is out of range low. However, there are a few tests that are graded by the degree to which the result is both out of range high and out of range low. Lab results that fall into this situation are considered “bi-directional”. For example, the lab test serum calcium can be toxic when the serum calcium result is out of range high, i.e. Hypercalcaemia and when the serum calcium result is out of range low, i.e. Hypocalcaemia. The categorical grading does not provide a distinction when the toxicity is out of range high versus out of range low. That is, the grading is still a positive number 1 thru 4 regardless of whether there is Hypercalcaemia or Hypocalcaemia.

Table 6 NCI CTCAE v3.0 Example Calcium Laboratory Criteria

CDISC Code (LBTESTCD)	CDISC Lab Name (LBTEST)	CDISC Lab Specimen Type LBSPEC	Preferred Unit of Measurement (LBCONU)	CTCAE v3.0 manual page number	Grade 1	Grade 2	Grade 3	Grade 4
CA	Calcium	serum	mg/dL	See "Calcium, serum-low", pg 40	<LLN - 8.0	<8.0 - 7.0	<7.0 - 6.0	<6.0
CA	Calcium	serum	mg/dL	See "Calcium, serum-high", pg 41	>ULN - 11.5	>11.5 - 12.5	>12.5 - 13.5	>13.5

As a consequence in the metadata, to account for the bi-directional derivation, the algorithm is entered into the metadata twice: once for out of range low results and once for out of range high results. In both instances, though the derivation resolves to the positive number grade regardless.

While applying the algorithm to “bi-directional” tests in the metadata is straightforward and the macro logic remains the same, when performing analysis and reporting one must take into consideration that a grade 3 result can vary across the same test depending on what toxicity is being described for bi-directional tests.

Normal Range Disparity

Our experience has shown that sometimes local lab vendor normal ranges are not in sync with the NCI CTCAE guidelines and the disparity can cause issues in the derivation of toxicity grade. Let’s reference the toxicity grade derivation for the lab test Glucose.

Table 7 Example Glucose Laboratory Criteria

	Grade 1	Grade 2	Grade 3	Grade 4
Glucose	>ULN – 160 mg/dL	>160 – 250 mg/dL	>250 – 500 mg/dL	>500 mg/dL

As you can see, a Grade 1 glucose result is a value that falls above the upper limit of normal to 160 mg/dL. If a particular lab vendor has an upper limit of normal set at 140 mg/dL, then a value that falls above 140 mg/dL and is less than 160 mg/dL will be assigned a toxicity grade 1 and subsequently more severe results from this same lab vendor will be assigned grades 2, 3, or 4, as appropriate.

On the other hand if a local lab vendor has an upper limit of normal set at 190 mg/dL, then the derivation is not so straightforward anymore. In this case the upper limit of normal is not below the expected Grade 1 result of 160 mg/dL but instead falls within the Grade 2 assignment range of glucose values. In this scenario, if a result is 165 mg/dL from this lab, then some decision must be made on the grading of this value. This value can be assigned a Grade 2 toxicity grade strictly per the NCI CTCAE guideline OR it can be assigned a NULL toxicity grade since the result falls within its local lab vendor normal range.

When this normal disparity occurs, the SAS macro described in this paper applies priority to the local lab result and its normal ranges. So that a result from the local lab that falls within it’s own lab normal range is assigned a NULL value despite the normal range being significantly wider than described in the NCI CTCAE guideline.

NCI CTCAE Version Updates

As noted earlier, one parameter of this SAS macro is the version number. This parameter was included in the design of the macro to enable a user to specify whether or not the macro should grade based on Version 3.0, Version 4.0 or some other version. The metadata provides an opportunity to enter the details of the guideline version into one location once and as such, it provides flexibility to the user to toggle between versions of the guideline.

CONCLUSION

A SAS Macro that applies the NCI CTCAE toxicity grading is a necessary and useful tool for analysis and reporting in Oncology trials. Although, the algorithms are well described in the NCI CCAE guidelines, there are several operational challenges to be aware of when implementing the guidelines globally on all lab results. However, despite the challenges of implementing the lab algorithms, a SAS macro can be developed with well-defined pre-requisites to accomplish this task.

REFERENCES

- Study Data Tabulation Model, Prepared by the CDISC Submission Data Standards Team, Version 1.1
SDTM Implementation Guide, Prepared by the CDISC Submission Data Standards Team, Version 3.1.2
Matthews, Carol. "Assigning NCI CTC Grades to Laboratory Results". 2010. Proceedings of PharmaSUG 2010 Conference. <http://www.lexjansen.com/pharmasug/2010/po/po03.pdf>
Common Terminology Criteria for Adverse Events v3.0. 2006.
http://ctep.cancer.gov/protocolDevelopment/electronic_applications/docs/ctcaev3.pdf

ACKNOWLEDGMENTS

The authors would like to thank the management team for their review of this paper and help to go through the company publication process.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Mei Dey
Enterprise: Accenture
City, State : Raleigh, NC
Work Phone: 267-305-6814
E-mail: mei.dey@accenture.com

Name: Lisa Pyle
Enterprise: Accenture
City, State : Philadelphia, PA
Work Phone: 267-305-6887
E-mail: elisabeth.pyle@accenture.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.